


EDITORALUTÉCIA

PROGRAMAS DE JOGOS DE TERROR



Para o Spectrum, TRS-80, TRS-80 Color, Apple, MSX, PC, Commodore 64,

VIC 20 e todos os micros nacionais compatíveis.

2ª EDIÇÃO

Título original inglês
WEIRD COMPUTER GAMES

Copyright © 1984 by Usborne Publishing Ltd.

Direitos de publicação exclusiva em língua portuguesa em todo o mundo
adquiridos pela

EDITORA LUTÉCIA LTDA.

Rua Argentina 171 — 20921 Rio de Janeiro, RJ — Tel.: 580-3668
que se reserva a propriedade literária desta tradução

Impresso no Brasil

PEDIDOS PELO REEMBOLSO POSTAL
Caixa Postal 23.052 — Rio de Janeiro, RJ — 20922

PROGRAMAS DE JOGOS DE TERROR

Jenny Tyler e Chris Oxlade

Tradução e Adaptação de Ronaldo Sergio de Biasi

LITEC

LIVRARIA EDITORA TÉCNICA LTDA

Rua dos Timbiras, 257 - CEP: 01208 - São Paulo

Tel.: 222-0477 - Fax: 220-2058

REF.

570

PREÇO

549,00

Sumário

- 2 Introdução
- 3 Torre do Terror
- 4 Os Crânios da Pirâmide
- 6 O Desafio dos Monstros
- 8 Tubarão
- 10 Receita de Bruxa
- 12 Micropuzzle
- 17 Respostas dos Problemas

EDITORALUTÉCIA

Introdução

Embora todos os micros nacionais utilizem a linguagem BASIC, existem vários dialetos desta linguagem usados em micros de diferentes marcas e modelos. Os micros nacionais podem ser divididos em seis "famílias": a família Spectrum (TK-85, TK-90X), a família TRS-80 (DGT-1000, CP-500 etc.), a família TRS-80 Color (CP-400), a família Apple (Micro Engenho, Ap-II etc.), a família MSX (Expert, Hot Bit) e a família PC (Ego, I-7000 etc.). Se você não souber a que família pertence o seu micro, poderá obter a informação em qualquer loja especializada.

Neste livro, a listagem principal de cada programa é para os micros da família TRS-80. As linhas que precisam ser mudadas para que o programa funcione em outros computadores estão indicadas por símbolos e as modificações aparecem no final de cada listagem. Os símbolos são os seguintes:

- família TRS-80 Color
- ▼ família Spectrum
- família Apple
- família MSX
- ◆ família PC
- ▲ Commodore 64 e VIC 20*

Cada vez que você vir o símbolo da família a que pertence o seu micro, procure no final da listagem uma linha com o mesmo símbolo e o mesmo número. Use-a no lugar da linha correspondente da listagem principal. Na página 17 há notas especiais para os usuários do TK-2000, do MSX e do PC.

*Até o momento não têm equivalentes no Brasil.

Como carregar e executar os programas

Escreva as linhas exatamente como estão na listagem, incluindo todos os espaços e sinais de pontuação. Depois de bater cada linha, verifique na tela se tudo está certo antes de apertar a tecla RETURN (ou a tecla equivalente para o seu micro).

Para começar o jogo, entre com RUN. Em alguns jogos, as coisas acontecem muito depressa, de modo que é melhor ler atentamente as instruções antes de começar. Se o programa não funcionar corretamente, o melhor é listá-lo e comparar as instruções com as do livro, uma por uma, até descobrir onde está o erro.

Experiências com os jogos

Este livro contém várias sugestões para modificar e ampliar os programas, mas não tenha medo de introduzir suas próprias alterações. Não há perigo de danificar o computador; se sua idéia não der certo, você sempre poderá voltar ao programa original.

Você provavelmente terá vontade de mudar a velocidade de alguns jogos, especialmente depois que estiver familiarizado com eles. As instruções para fazer isso estão perto da listagem de cada programa.

Se o seu computador é capaz de gerar cores ou sons, consulte o manual para descobrir como funcionam as instruções correspondentes e experimente usá-las nos programas deste livro.

Ilustrações: Rob McCaig, Sue Walliker, Martin Newton e Graham Smith.

Diagramação: Roger Priddy.

Programa "Micropuzzle": Les Howarth.

Programa "O Desafio dos Monstros": Adrian Hall.

Torre do Terror



Você sente a garganta seca, as pernas trêmulas, um frio na espinha... está entrando na Torre do Terror!

Aperte A para andar pelos quartos. Oh! Um esqueleto! Logo adiante, um fantasma. Não! Até um lobisomem! A cada novo susto, seu coração bate mais depressa.

Você pode escolher entre avançar (A) ou recuar (R) para se acalmar um pouco. Não demore demais... só tem até a meia-noite para chegar ao terraço e conquistar o Tesouro. Por outro lado, cuidado com o ritmo cardíaco. Se passar de 150, você ficará louco e nada poderá impedi-lo de pular pela janela.

Experimente mudar o limite dos batimentos cardíacos.

Você sabe como aumentar o número de quartos?

Veja se é capaz de incluir outros tipos de monstros no programa.

Qual é o Tesouro da Torre? Acrescente uma mensagem ao programa para explicar isso.

Como funciona

- 10: Vai para a parte onde são lidos os dados.
- 30: RM é o quarto onde você está.
- 40-50: Escolhe a hora inicial e fixa o ritmo cardíaco.
- 90: Imprime o comentário armazenado em RS.
- 100-160: Verifica em que andar e quarto você está e imprime esta informação.
- 170-180: Imprime outras informações.
- 190: GF é uma bandeira que é levantada quando há algo assustador no quarto seguinte.
- 200: Verifica se você chegou ao terraço.
- 210: Decide em que quarto vai ficar o alcapão.
- 220: Decide se vai haver algo assustador no quarto seguinte. Em caso afirmativo, vai para uma sub-rotina.
- 230-280: Recebe uma instrução do jogador e age de acordo com ela.
- 290: Incrementa o tempo.
- 300: Verifica se já é meia-noite.
- 310: Verifica se o ritmo cardíaco passou de 150.
- 330: Verifica se você está no quarto onde se encontra o alcapão.
- 400-410: Sub-rotina para avançar. Aumenta o ritmo cardíaco se há algo assustador no quarto.
- 420-460: Escolhe o tipo de monstro e o grau do seu susto.
- 470-510: Leitura dos dados (inicialização).
- 520: Sub-rotina do alcapão.

Modificações

- ◆ 40, 210, 220, 290, 330, 420, 440 Substitua RND(0) por RND
- ▲ 40, 210, 220, 290, 330, 420, 440 Substitua RND(0) por RND(1)
- ▲ 60 PRINT CHR\$(147):PRINT:PRINT
- 60 HOME:PRINT:PRINT
- ▲ 240 GET IS
- 240 IS="" IF PEEK(-16384))127 THEN GET IS
- ▼ 470 DIM GS(3,15)

```

10 GOSUB 470
20 LET RS="BOA SORTE"
30 LET RM=0
40 LET H=21:LET M=INT(RND(0)*10)+10
50 LET P=50
60 CLS:PRINT:PRINT
70 PRINT "TORRE DO TERROR"
80 PRINT "=====
90 PRINT:PRINT RS
100 LET RS="":LET FL=INT(RM/5)
110 LET R=RM-FL*5+1
120 PRINT:PRINT"VOCE ESTA NO";
130 IF FL=0 THEN PRINT " ANDAR TERREO"
140 IF FL=6 THEN PRINT " TERRACO"
150 IF FL>0 AND FL<6 THEN PRINT " ANDAR ";FL
160 IF FL<6 THEN PRINT "NO QUARTO NUMERO ";R
170 PRINT:PRINT "SAO ";H;" HORAS E ";M;" MINUTOS"
180 PRINT:PRINT "SEU RITMO CARDIACO E ";P
190 LET GF=0
200 IF RM=30 THEN GOTO 350
210 LET TR=INT(RND(0)*9+1)
220 IF RND(0)>.6 THEN GOSUB 420
230 PRINT:PRINT "RECUAR OU AVANCAR (R/A)?"
240 LET IS=INKEY$
250 IF IS("<")="A" AND IS(">")="R" THEN GOTO 240
260 IF IS="A" THEN GOSUB 400
270 IF IS="R" THEN LET RM=RM-1:LET P=P-5
280 IF RM=1 THEN LET RM=0
290 LET M=M+INT(RND(0)*3+1):IF M>59 THEN M=M-60:LET H=H+1
300 IF H=24 THEN GOTO 360
310 IF P>150 THEN GOTO 380
320 IF P<40 THEN LET P=40
330 IF FL=TR AND RND(0)>.5 THEN GOSUB 520
340 GOTO 60
350 PRINT "MUITO BEM ":STOP
360 PRINT:PRINT "E MEIA NOITE!"
370 PRINT:PRINT "TARDE DEMAIS!":STOP
380 PRINT "VOCE FICOU LOUCO E"
390 PRINT "SE JOGOU POR UMA JANELA":STOP
400 IF GF=1 THEN LET P=P+S*2:LET RS="AAAAHHH!"
410 LET P=P-1:LET RM=RM+1:RETURN
420 LET TY=INT(RND(0)*3+1)
430 LET US=GS(TY)
440 LET S=INT(RND(0)*5)+FL+TY*2
450 PRINT:PRINT "A FRENTE VOCE VE UM ";US
460 LET GF=1:RETURN
470 DIM GS(3)
480 LET GS(1)="ESQUELETO"
490 LET GS(2)="FANTASHA"
500 LET GS(3)="LOBISOMEM"
510 RETURN
520 LET RS="VOCE CAIU EM UM ALCAPAO!":LET RM=RM-5:LET P=P+10:RETURN
    
```


Os Crânios da Pirâmide

Uma mente sem corpo... foi assim que você ficou depois daquele terrível acidente que sofreu durante a construção da Grande Pirâmide.

Você precisa de toda a sua energia mental para satisfazer ao Espírito da Pirâmide. Todo dia, ele lhe fornece um número. Você deve fazer com que cinco crânios caiam do topo da Grande Pirâmide em pedras numeradas, de forma a somarem aquele número.

Aperte 1, 2, 3 ou 4 para soltar um crânio. Obtenha o número pedido e estará livre da Maldição da Pirâmide.



Veja se é capaz de substituir o símbolo do crânio por um símbolo gráfico.

```
10 LET TS=0:LET IS="0":LET SP=3
20 LET N=INT(RND(0)*30)+10
30 CLS:PRINT
40 PRINT "OS CRANIOS DA PIRAMIDE"
50 PRINT
60 PRINT "VOCE PRECISA"
70 PRINT "OBTER ";N:PRINT
80 LET K=3000:GOSUB 540
90 CLS:LET Y=2
100 FOR I=1 TO 4
110 LET AS=CHR$(48+I):LET X=6+2*I
120 GOSUB 560
130 NEXT I
140 LET AS="↑":FOR I=1 TO 5
150 LET Y=2+2*I
160 FOR J=1 TO I+3
170 LET X=7-I+2*J
180 GOSUB 560
190 NEXT J
200 NEXT I
210 LET Y=14
220 FOR I=1 TO 9:LET AS=CHR$(48+I)
230 LET X=1+2*I:GOSUB 560
240 NEXT I
250 LET S=0
260 FOR I=1 TO 5
270 LET AS="ESCOLHA COLUNA"
280 LET X=0:LET Y=0:GOSUB 560
290 LET IS=INKEY$
300 IF IS="" THEN GOTO 290
310 LET R=VAL(IS)
320 IF R<1 OR R>4 THEN GOTO 290
330 LET Y=3:LET K=400:LET X=6+2*R
340 LET F=0
```

Como o programa funciona

10: Dá valores iniciais às variáveis.

20: Escolhe o número que o jogador deve obter.

30-80: Imprime mensagem inicial e espera para que o jogador possa vê-la.

90-240: Imprime a pirâmide e os números no topo e na base da pirâmide.

250: Inicia a contagem de pontos em zero.



Se o seu micro não tem o símbolo ↑, substitua-o por ! na linha 140.

Aumente o valor do SP na linha 10 para acelerar o programa

260: Fixa o número de crânios.

270-280: Imprime mensagem no canto superior esquerdo da tela.

290-300: Espera que uma tecla seja apertada.

310-320: Verifica se a tecla apertada está entre 1 e 4. Se não está, volta para 290.

330: X e Y são as coordenadas do crânio. Nesta linha, o crânio é colocado no topo da pirâmide.

340: F é uma bandeira que assinala se o movimento anterior foi um desvio lateral.

Veja se consegue
aumentar o número de
crânios em cada jogada.



```

350 LET AS=""
360 GOSUB 560
370 GOSUB 540
380 LET AS="" :GOSUB 560
◆▼▲● 390 IF F=0 AND Y<13 THEN LET X=X+SGN(RND(0)-.5)
400 LET Y=Y+1:LET F=1-F
410 IF Y<14 THEN GOTO 350
420 LET S=S+(X-1)/2
430 LET X=15:LET Y=0
440 LET AS=STR$(I)+" "+STR$(S)
450 GOSUB 560
460 NEXT I
470 LET K=1000:GOSUB 540
480 LET D=ABS(S-N):LET TS=TS+D
▲● 490 CLS:PRINT:PRINT
500 PRINT "SUA CONTAGEM AGORA E ";TS
510 IF D>1 THEN LET K=3000:GOSUB 540:GOTO 20
520 PRINT "E VOCE ESTA LIVRE PARA IR EMBORA!"
530 STOP
540 FOR T=1 TO K STEP SP:NEXT T
550 RETURN
○◆▼▲● 560 PRINT@ Y*64+X,AS;
570 RETURN

```

350-370: Imprime crânio e espera.

380: Remove o crânio.

390: Escolhe o sentido do desvio
(se o valor de F permite um desvio
lateral).

400: Faz o crânio descer uma linha.

410: Verifica se o crânio chegou à
base da pirâmide.

420: Calcula a soma de acordo com
a coluna em que o crânio termina.

430-450: Imprime a soma.

470: Espera.


480-500: Calcula a diferença entre a
soma e o número desejado.
Acrescenta essa diferença à
contagem total. Limpa a tela e
imprime a contagem.

510: Se a diferença é maior que 1,
volta para uma nova tentativa.

520: Imprime mensagem final.

540-550: Sub-rotina para esperar
por um tempo que depende de K.


560-570: Sub-rotina para colocar o
cursor na posição X,Y da tela.



Você seria
capaz de mudar
os limites do
número que o
jogador deve
obter?

Modificações

- ▲● 20,390 Substitua RND(0) por RND(1)
- ◆▼ 20,390 Substitua RND(0) por RND
- 30,90,490 Substitua CLS por HOME
- ▲ 30,90,490 Substitua CLS por PRINT CHR\$(147)
- 290 IS="":IF PEEK(-16384)>127 THEN GET IS
- ▲ 290 GET IS
- ▼ 560 PRINT AT Y,X;AS
- 560 VTAB(Y+1):HTAB(X+1):PRINT AS
- ▲ 560 PRINT CHR\$(19);:FOR LL=1 TO Y:
PRINT:NEXT:PRINT TAB(X);AS
- ◆■ 560 LOCATE X,Y:PRINT AS
- 560 PRINT@ Y*32+X,AS;



O Desafio dos Monstros

Lutar com monstros é um esporte perigoso! Pior para você, que se prepara para enfrentar alguns dos monstros mais ferozes do Universo. Para isso, terá que usar toda a sua agilidade mental. Deverá calcular, por exemplo, o esforço muscular necessário para deter o monstro, o que implica multiplicar o tamanho do monstro pela distância a que ele se encontra de você.

Se os números parecerem muito difíceis, você pode apertar o Botão de Pânico (tecla P). Nesse caso, terá que calcular a quantidade de adrenalina de que o seu corpo necessita para sobreviver à crise, dividindo o aumento do ritmo cardíaco pelo consumo de oxigênio. Cuidado, porém, para não usar demais o Botão de Pânico, ou poderá forçar o coração e perder os sentidos.

Para vencer o desafio, você terá que sobreviver a 12 rounds, cada um contra um monstro diferente.

Como o programa funciona

20: Faz o número de vezes que o Botão de Pânico foi usado igual a zero.

30-60: Estabelece limites para o tamanho e a distância do monstro.

70-75: N é o número de rounds.

80: Escolhe o tamanho do monstro.

90: Escolhe a distância.

100-120: Aumenta de 0,5 os limites de tamanho e distância para o round seguinte.

210-220: Pede uma resposta e vai para uma sub-rotina para recebê-la.

230: Se a resposta está errada, pula para 320.

250: Imprime mensagem se a resposta está certa.

260: Volta para o round seguinte.

270-310: Imprime mensagem se você sobreviveu a 12 rounds.

330-380: Imprime mensagem se você errou.

Se achar o jogo muito difícil, torne-o mais lento aumentando o valor de Q na linha 660.



Se acha o jogo fácil demais, diminua Q.

```
20 LET P=0
30 LET K=3
40 CLS
50 LET X=1
60 LET Y=6
70 LET N=-1
75 LET N=N+1
80 LET G=INT(RND(0)*Y+X)
90 LET I=INT(RND(0)*K+K)
100 LET Y=Y+0.5
110 LET X=X+0.5
120 LET K=K+0.5
130 PRINT
140 PRINT
150 PRINT "TAMANHO DO MONSTRO: ";
160 PRINT G
170 PRINT
180 PRINT "DISTANCIA: ";
190 PRINT I
200 PRINT
210 PRINT "ESFORÇO MUSCULAR? ";
220 GOSUB 570
230 IF Z(>G*I THEN GOTO 320
240 CLS
250 PRINT "MONSTRO SOB CONTROLE"
260 IF N<11 THEN GOTO 75
270 PRINT "UFA!!! O MONSTRO"
280 PRINT "ESTA CANSADO E FOI"
290 PRINT "PROCURAR OUTRA VITIMA."
300 PRINT "VOCE ESCAPOU!!!"
310 STOP
320 CLS
330 PRINT "VOCE FOI ESMAGADO"
340 PRINT "ATE A MORTE"
350 PRINT "PELO MONSTRO"
360 PRINT
370 PRINT "VOCE SOBREVIVEU ";N;" ROUNDS"
```




```

380 STOP
▲●390 CLS
▼▼400 LET WX=INT(RND(0)*9+1)
▼▼410 LET WY=INT(RND(0)*9+1)
420 LET W=WX*WY
430 LET P=P+1
440 IF P=4 THEN GOTO 700
450 IF P=3 THEN PRINT "ESTA VENDO ESTRELAS"
460 PRINT "PANICO!!!"
470 PRINT
480 PRINT "RITMO CARDIACO: ";W
490 PRINT "CONSUMO DE OXIGENIO: ";WX
500 PRINT
510 PRINT "QUANTIDADE DE ADRENALINA? ";
520 LET Q=100
530 GOSUB 580
540 IF Z(>)WY THEN GOTO 320
▲●550 CLS:IF N(11).THEN GOTO 75


```

Torne o jogo mais difícil aumentando o número de rounds.

```

560 GOTO 270
570 LET Q=0
▼580 LET Z$=""
▲●590 LET AS=INKEY$
▼▼600 IF AS=CHR$(13) THEN GOTO 680
▼610 IF AS="P" THEN GOTO 390
620 IF VAL(AS)=0 AND AS(>)"0" THEN GOTO 650
▼630 PRINT AS;
640 LET Z$=Z$+AS
650 LET Q=Q+1
▼660 IF Q=500 THEN GOTO 320
670 GOTO 590
680 LET Z=VAL(Z$)
690 RETURN
▲●700 CLS
710 PRINT "VOCE DESMAIOU"
720 STOP

```



400-410: Escolhe o consumo de oxigênio e a quantidade de adrenalina.
 420: Escolhe o aumento do ritmo cardíaco.
 430-450: Incrementa a variável do pânico, imprime advertência quando chega a 3 e encerra o jogo quando chega a 4.
 460-510: Imprime mensagens e pede resposta.
 520: Fixa o valor de Q no início da contagem das linhas 650-660.
 530: Vai para a sub-rotina que analisa as respostas.
 540: Se a resposta está errada, vai para 320.
 550-560: Se a resposta está certa, encerra o jogo ou volta para o round seguinte.
 570-690: SUB-ROTINA PARA ANALISAR AS RESPOSTAS
 570: Faz o contador de tempo igual a zero.
 580: Reserva memória para Z\$.
 590: Verifica qual a tecla que foi apertada.
 600: Verifica se a tecla apertada foi RETURN [CHR\$(13)]. Em caso afirmativo, pula para 680.
 610: Verifica se a tecla P foi apertada. Em caso afirmativo, vai para a rotina de pânico.
 620: Se nenhuma tecla foi apertada, pula para 650 para aumentar o valor de Q.
 630: Se outra tecla que não RETURN ou P foi apertada, imprime o caractere correspondente.
 640: Z\$ contém todas as teclas que foram apertadas nesse round.
 650-660: Contagem de tempo. Se Q chega a 500 sem que nenhuma tecla tenha sido apertada, vai para 320.
 680: Transforma os algarismos da "string" Z\$ em um valor numérico, para poder compará-lo com G*I na linha 230.
 710: Mensagem que aparece quando você usa o Botão de Pânico pela quarta vez.

Modificações

- ▲ 40,240,320,390,550,700 Substitua CLS por PRINT CHR\$(147)
- 40,240,320,390,550,700 Substitua CLS por HOME
- ▲● 80,90,400,410 Substitua RND(0) por RND(1)
- ▼ 80,90,400,410 Substitua RND(0) por RND
- ▼ 580 LET Z\$=""
- ▲ 590 GET AS
- 590 AS="" IF PEEK(-16384))127 THEN GET AS
- ▼ 595 IF AS="" THEN GOTO 650
- ▼ 597 PAUSE 40
- ▼ 600 IF AS=CHR\$(13) THEN PAUSE 20:GOTO 680
- ▼ 610 IF AS="P" THEN PAUSE 30:GOTO 390
- ▼ 630 PRINT AS;:PAUSE 15
- ▼ 660 IF Q=250 THEN GOTO 320

Tubarão

Hummmmm... as pessoas são tão gostosas! Acontece que, com o caçador sempre no seu encalço, não é sempre que você consegue apanhá-las.

Você é o T (tubarão) e pode se movimentar apertando as teclas A, Z, N e M. Veja quantos P (pessoas) consegue pegar antes de ser alcançado pelo C (caçador). Naturalmente, existe um senão. Cada vez que você consegue apanhar e devorar um delicioso ser humano, fica tão satisfeito que não consegue mais se lembrar da direção associada a cada tecla.

```

▼ 10 DIM M$(12):DIM K$(4):GOSUB 660
  20 LET S=0:LET T=0:LET G=0
  30 LET U=1
  40 LET K$(1)="A":LET K$(2)="M"
  50 LET K$(3)="Z":LET K$(4)="N"
  60 LET PX=2:LET PY=2
  70 GOSUB 600
  80 GOSUB 490:GOSUB 440
  90 LET NX=PX:LET NY=PY
  ▲ 100 LET IS=INKEY$
  110 IF IS=K$(1) THEN LET NY=NY-1
  120 IF IS=K$(2) THEN LET NX=NX+1
  130 IF IS=K$(3) THEN LET NY=NY+1
  140 IF IS=K$(4) THEN LET NX=NX-1
  150 LET X=NX:LET Y=NY:GOSUB 530
  160 IF F=1 THEN GOTO 230
  170 LET X=PX:LET Y=PY:LET AS=" "
  180 GOSUB 640
  190 LET X=NX:LET Y=NY:LET AS="T"
  200 GOSUB 640
  210 LET PX=NX:LET PY=NY
  220 IF PX=TX AND PY=TY THEN GOSUB 340
  230 IF PX=GX AND PY=GY THEN GOTO 280
  ◆▼▲● 240 IF RND(0)>U THEN GOSUB 370
  250 LET X=14:LET Y=12:LET AS=STR$(T)
  260 GOSUB 640
  270 LET G=G+1:GOTO 90
  280 FOR P=1 TO 2000:NEXT P
  
```

Como faria para o caçador se mover mais frequentemente?

Modificações

```

▼ 10 DIM M$(12,16):DIM K$(4):GOSUB 660
● 100 IS="":IF PEEK(-16284))127 THEN GET IS
▲ 100 GET IS
▲■ 240,390,400,550,560 Substitua RND(0) por RND(1)
◆▼ 240,390,400,550,560 Substitua RND(0) por RND
● 290,600 Substitua CLS por HOME
▲ 290,600 Substitua CLS por PRINT CHR$(147)
▼ 530 LET F=0:IF M$(Y,X)<>" " THEN LET F=1
▼ 640 PRINT AT Y,X;AS
● 640 VTAB(Y):HTAB(X):PRINT AS
▲ 640 PRINT CHR$(19);:FOR LL=1 TO Y:PRINT:PRINT TAB(X);AS
○ 640 PRINT Y*32+X,AS;
◆■ 640 LOCATE X,Y:PRINT AS
  
```

Como o programa funciona

10: Reserva espaço na memória para o mapa. Vai para uma sub-rotina onde são lidos os dados do mapa.

20: Dá valores iniciais às variáveis.

30: Estabelece a frequência com que o caçador se movimenta.

40-50: Estabelece as direções correspondentes às teclas.

60: Estabelece a posição inicial do tubarão.

70: Vai para a sub-rotina que imprime o mapa.

80: Vai para sub-rotinas que escolhem as posições do caçador e das pessoas.

90: Estabelece uma nova posição para o tubarão.

100-140: Calcula uma nova posição para o tubarão, dependendo da tecla que foi apertada.

150: Verifica se a nova posição coincide com um obstáculo.

160: Em caso afirmativo, o tubarão não se move.

170-180: Imprime um espaço na posição antiga do tubarão.

190-200: Imprime um T na nova posição do tubarão.

210: Muda a posição do tubarão.

220: Verifica se a nova posição coincide com um P; em caso afirmativo, vai para uma sub-rotina.



230: Verifica se a nova posição coincide com o C; em caso afirmativo, pula para 280 para terminar o jogo.

240: Movimenta o caçador.

250-260: Imprime a contagem de pontos até o momento.

270: Incrementa o número de movimentos e volta para o movimento seguinte.

280: Faz uma pausa para que você possa ver se foi apanhado pelo caçador.





```

▲● 290 CLS:PRINT "VOCE FOI APANHADO!"
300 LET S=INT(10000*(T/6)):PRINT "VOCE FEZ ";S;" PONTOS"
310 PRINT:PRINT "QUER JOGAR DE NOVO? (S/N)"
320 INPUT AS:IF AS="S" THEN GOTO 20
330 STOP
340 LET T=T+1:LET U=U-0.03
350 GOSUB 390:GOSUB 490
360 RETURN
370 LET X=GX:LET Y=GY:LET AS=" "
380 GOSUB 640:GOSUB 440:RETURN
◆▼▲● 390 LET K1=INT(RND(0)*4)+1
◆▼▲● 400 LET K2=INT(RND(0)*4)+1
410 LET TS=KS(K1)
420 LET KS(K1)=KS(K2):LET KS(K2)=TS
430 RETURN
440 GOSUB 550
450 IF X=TX AND Y=TY THEN GOTO 440
460 LET GX=X:LET GY=Y
470 LET AS="C":GOSUB 640
480 RETURN
490 GOSUB 550
500 LET TX=X:LET TY=Y
510 LET AS="P":GOSUB 640
520 RETURN
▼ 530 LET F=0:IF MID$(MS(Y),X,1)<>" " THEN LET F=1
540 RETURN
◆▼▲● 550 LET X=INT(RND(0)*14)+2
◆▼▲● 560 LET Y=INT(RND(0)*9)+2
570 GOSUB 530
580 IF F=1 THEN GOTO 550
590 RETURN
▲● 600 CLS:LET X=1
610 FOR Y=1 TO 12:LET AS=MS(Y)
620 GOSUB 640:NEXT Y
630 RETURN
○◆▼▲● 640 PRINT# Y*64+X,AS;
650 RETURN
660 FOR I=1 TO 12:READ MS(I):NEXT I
670 RETURN
680 DATA "0000000000000000"
690 DATA "0 0"
700 DATA "0 0 000 00 000 0"
710 DATA "0 0 0 0"
720 DATA "0 0 0000 0 0 0"
730 DATA "000 0 0 0 0 0"
740 DATA "0 00 0"
750 DATA "0 0 00 0 0 0 0"
760 DATA "0 0 00 0 000 0 0"
770 DATA "0 0"
780 DATA "0000000000000000"
790 DATA "PESS.COMIDAS:"

```

290-330: Imprime mensagem e contagem de pontos se o tubarão é apanhado. Pergunta se você quer jogar de novo.



A contagem de pontos depende do número de pessoas comidas e do número de movimentos.

340-360: Sub-rotina que aumenta a contagem e a probabilidade de o caçador mudar de posição e vai para outras sub-rotinas para trocar a função das teclas e mudar a posição da pessoa.

370-380: Sub-rotina que retira o caçador da antiga posição e o imprime em uma nova posição.

390-430: Sub-rotina para trocar a função das teclas.

440-480: Sub-rotina para imprimir o caçador, usando outra rotina para escolher uma posição para ele. A linha 450 verifica se a posição escolhida já está ocupada pela pessoa.

490-520: Sub-rotina para imprimir a pessoa, usando outra sub-rotina para escolher sua posição.

530-540: Sub-rotina que verifica se a posição X,Y coincide com um obstáculo. Em caso afirmativo, F assume o valor 1.

550-590: Sub-rotina que escolhe uma posição ao acaso e verifica se coincide com um obstáculo.

600-630: Sub-rotina para imprimir o mapa.

640-650: Sub-rotina para imprimir na posição X,Y.

660-790: Sub-rotina para ler os dados do mapa.



Estas linhas contêm os dados para o mapa. Para mudá-las, desenhe um novo mapa em papel quadriculado e copie-o no programa, usando 0 para os obstáculos e espaços para as passagens.



Receita de Bruxa



Como assistente da Superbruxa, sua missão é percorrer o campo em uma vassoura voadora, recolhendo ingredientes para as poções maléficas da feiticeira. Você sabe exatamente o que procurar, porque ela transmite as receitas diretamente para a tela do seu computador.

Ingredientes como U (Unha de Morcego) e O (Olho de Lagarto) passam por baixo de você. Aperte qualquer tecla para descer e apanhá-los (para ser bem-sucedido, você tem que descer bem em cima da letra). Tome cuidado, porque um deles pode ser o Ladrão de Vassouras disfarçado. Se você colidir com o ladrão, perderá a vassoura. (Felizmente, você começa o jogo com três vassouras de reserva.) Se não conseguir todos os ingredientes, a Superbruxa não vai ficar nada satisfeita...

```

5 CLEAR 200
10 GOSUB 700:GOSUB 620
20 LET NW=4:LET M=0:LET FM=50
30 GOSUB 530:LET H=NH
40 GOSUB 550
50 LET EF=0:LET FF=0:LET HH=H
60 FOR I=1 TO 4:LET G(I)=0:NEXT I
70 FOR I=1 TO 4:LET AS=PS(I)+":":LET X=1
80 LET Y=1+I:GOSUB 610:NEXT I
90 GOSUB 490:GOSUB 430
100 IF M=FM THEN GOTO 250
110 IF EF=1 THEN GOTO 230
120 IF INKEY$="" THEN GOTO 90
130 LET HH=H-1:GOSUB 510
140 GOSUB 490:GOSUB 430
150 LET H=HH
160 IF H>0 THEN GOTO 130
170 IF P=17 THEN GOSUB 320
180 IF FF=1 THEN GOTO 240
190 GOSUB 530
200 LET HH=H+1:GOSUB 510:GOSUB 490
210 LET H=HH:IF H=NH THEN GOTO 90
220 GOSUB 430:GOTO 200
230 LET AS="MUITO BEM! A SUPERBRUXA FICOU SATISFEITA!"
240 LET AS="ACABARAM AS VASSOURAS!":GOTO 260
250 LET AS="TARDE DEMAIS!"
260 LET X=1:LET Y=15:GOSUB 610
270 STOP
280 LET AS="W":LET X=P:GOSUB 610
290 FOR T=1 TO 1000:NEXT T
300 LET NW=NW-1:IF NW=0 THEN LET FF=1
310 RETURN
320 IF W=1 THEN GOSUB 280:RETURN
330 LET G(R)=G(R)+1
340 LET AS=STR$(G(R)):LET X=17
350 LET Y=1+R:GOSUB 610
360 LET P=19:GOSUB 430
370 LET F=0
380 FOR I=1 TO 4
390 IF G(I)>N(I) THEN LET F=F+1

```

Como o programa funciona

20: Inicializa as variáveis (veja se descobre o que representam).
30: H é a sua altura acima do solo. NH é a nova altura depois de um mergulho.

50: EF e FF são bandeiras que são levantadas quando o jogo termina. HH é uma variável para guardar alturas intermediárias.

60: A matriz G é usada para contar quantos objetos de cada tipo você já apanhou.

70-80: Imprime a receita.

90-120: Faz o "solo" e os objetos se moverem na tela até uma tecla ser apertada.

130-160: Mantém o solo e os objetos em movimento e faz a bruxa descer até o solo.

170: Se P é 17, é porque a bruxa pegou um objeto.

180: Verifica se o jogo terminou.

200-220: Mantém o solo e os objetos em movimento e faz a bruxa subir de novo.


230-270: Mensagens finais.

280-310: Se o objeto é o Ladrão de Vassouras disfarçado, esta sub-rotina imprime W na tela, diminui NW e levanta FF se as vassouras acabaram.


320: Sub-rotina que verifica se o objeto é o Ladrão de Vassouras.

330-350: Aumenta de 1 o número de objetos do tipo R recolhidos e mostra o novo total na tela.

370-420: Verifica se todos os ingredientes da receita foram recolhidos.



Está achando difícil? Torne o jogo mais fácil aumentando o valor de FM na linha 20, o que aumenta o número total de objetos.



Você pode substituir as letras por símbolos gráficos.

Experimente mudar os ingredientes.



```

400 NEXT I
410 IF F=4 THEN LET EF=1
420 RETURN
430 LET GD=GD+1:IF GD=6 THEN LET GD=1
440 LET AS=GS(GD):LET X=1:LET Y=15:GOSUB 610
450 LET AS=" ":LET X=P:LET Y=14:GOSUB 610
460 LET P=P+1:IF P=20 THEN GOSUB 550
470 LET X=P:LET AS=CS:GOSUB 610
480 RETURN
490 LET AS="-Y=":LET X=16:LET Y=14:HH
500 GOSUB 610:RETURN
510 LET AS=" ":LET X=16:LET Y=14:H
520 GOSUB 610:RETURN
◆▼▲■● 530 LET NH=INT(RND(0)*5+3)
540 RETURN
◆▼▲■● 550 LET R=INT(RND(0)*4)+1
560 LET W=0:IF RND(0)>0.8 THEN LET W=1
570 LET CS=LEFT$(PS(R),1):LET P=1
580 LET M=M+1
590 IF M=FM THEN LET CS="CALDEIRAO"
600 RETURN
◆▼▲■● 610 PRINT@ Y*64+X,AS:RETURN
▲● 620 CLS:PRINT:PRINT
630 PRINT "A POCAO DEVE TER":PRINT
640 FOR I=1 TO 4
◆▼▲■● 650 LET N=INT(RND(0)*3)+2:LET N(I)=N
660 PRINT:PRINT:N;" ";PS(I)
670 NEXT I:PRINT:PRINT
680 PRINT "APORTE RETURN PARA JOGAR"
▲● 690 INPUT XS:CLS:RETURN
▼ 700 DIM PS(4):DIM N(4):DIM G(4)
▼ 710 DIM GS(5)
720 FOR I=1 TO 4:READ PS(I):NEXT I
730 DATA "UNHAS DE MORCEGO","OLHOS DE LAGARTO","RAIZES DE URTIGA","LINGUAS DE SAPO"
740 LET HS="")====>====>====>====>===="
▼ 750 FOR I=1 TO 5:LET GS(6-I)=MID$(HS,I,1+20)
760 NEXT I:LET GD=1
770 RETURN

```



430-480: Imprime um novo trecho do solo e move o objeto até a posição P. Se o objeto passa pela bruxa (P=20), escolhe um novo objeto.

490-500: Imprime a bruxa.

510-520: Apaga a bruxa.

530-540: Escolhe uma nova altura para a bruxa.

550: Escolhe um objeto.

560: Decide se o objeto vai ser o Ladrão de Vassouras.

570: Coloca a primeira letra do objeto em CS e a faz posição do objeto (P) igual a 1.

580: Incrementa M, o número total de objetos.

590: Se M é igual ao número máximo de objetos permitido, CS é mudado para "CALDEIRÃO".

610: Sub-rotina geral para posicionar o cursor e imprimir mensagens.

620-690: Escolhe e imprime a receita.

700-770: Inicialização do programa. Os nomes dos objetos são guardados em PS e os símbolos do solo em HS. A linha 750 faz com que diferentes partes do solo sejam guardadas nos elementos da matriz GS.

Que fará a Superbruxa se você perder? Acrescente outros comentários ao jogo.



Modificações

○◆▼▲■● Suprima a linha 5

▲ 120 GET IS:IF IS="" THEN GOTO 90

● 120 IS="" IF PEEK(-16384)>127 THEN GET IS

● 125 IF IS="" THEN GOTO 90

▲■● 530,550,560,650 Substitua RND(0) por RND(1)

◆▼ 530,550,560,650 Substitua RND(0) por RND

▼ 570 LET CS=PS(R,1):LET P=1

▼ 610 PRINT AT Y,X,AS:RETURN

▲ 610 PRINT CHR\$(19):FOR LL=1 TO Y:PRINT: NEXT:PRINT TAB(X),AS:RETURN

● 610 VTAB(Y)=HTAB(X):PRINT AS:RETURN

○ 610 PRINT@ Y*32+X,AS:RETURN

◆■ 610 LOCATE X,Y:PRINT AS

▲ 620,690 Substitua CLS por PRINT CHR\$(147)

● 620,690 Substitua CLS por HOME

▼ 700 DIM PS(4,15):DIM N(4):DIM G(4)

▼ 710 DIM GS(5,24)

▼ 750 FOR I=1 TO 5:LET GS(6-I)=HS(I TO I+20)



Micropuzzle

Que aconteceu? Onde está você? Tudo parece normal, embora você se sinta um pouco tonto. É melhor olhar em torno e ver se consegue descobrir o que está acontecendo. (Pista: Converse com o computador usando as palavras que estão nas linhas 1810 a 1880. Para se movimentar, você também pode usar as abreviações L=ir para leste, C=ir para cima etc.)

Este é um jogo de aventura. Você descobrirá o que fazer durante o próprio jogo.



Como o programa funciona

10 CLEAR 300

20 GOSUB 1640

20: Vai para a sub-rotina onde são lidos todos os dados.

30 RESTORE:FOR I=1 TO R:READ D\$:NEXT I

40 CLS:PRINT "MICRO PUZZLE"

50 PRINT "=====

60 PRINT "VOCE ESTA ";D\$

30-60: Imprime a descrição de um quarto.

70 IF R=20 AND F(9)=0 THEN PRINT "VOCE VE UM GATO ENORME"

80 FOR I=1 TO 6

90 IF L(I)=R AND F(I)=0 THEN PRINT "HA UM";:IF LEFT\$(D\$(I),1)="A" THEN PRINT D\$(I);" AQUI.":ZZ=1

100 IF L(I)=R AND F(I)=0 AND ZZ=0 THEN PRINT RIGHT\$(D\$(I),LEN(D\$(I))-1);" AQUI."

110 ZZ=0:NEXT I

80-110: Verifica se há algum objeto no quarto.

120 PRINT:PRINT "VOCE PODE IR PARA ";

130 FOR I=1 TO LEN(R\$(R))

140 PRINT MID\$(R\$(R),I,1);:IF I<LEN(R\$(R)) THEN PRINT ",";

150 NEXT I

120-150: Imprime as direções para onde o jogador pode ir.

160 PRINT:PRINT "-----":PRINT M\$:LET M\$="HEIN?"

170 IF F(16)=1 THEN PRINT "CONTAGEM REGRESSIVA PARA AUTO-DESTRUCAO EM ";L

180 PRINT "QUE VAI FAZER AGORA":LET Q\$="":INPUT Q\$

190 LET V\$="":LET W\$="":LET VB=0:LET OB=0:LET LI=LEN(Q\$)

200 FOR I=1 TO LI

210 IF MID\$(Q\$,I,1)=" " AND V\$="" THEN LET V\$=LEFT\$(Q\$,I-1)

220 IF MID\$(Q\$,I+1,1)<>" " AND V\$<>" " THEN LET W\$=RIGHT\$(Q\$,LI-I):LET I=LI

230 NEXT I

180-230: Recebe as instruções do jogador e as divide em verbo e objeto.


```

240 IF W$="" THEN LET V$=Q$
250 FOR I=1 TO V:IF V$=H$(I) THEN LET VB=I
260 NEXT I

```

240-260: Verifica se a primeira palavra é um verbo conhecido.

```
270 GOSUB 430
```

270: Vai para uma sub-rotina que examina o resto da instrução.

```

280 IF VB=0 THEN LET VB=V+1
290 IF Q$="" THEN LET M$="DIGA ALGUMA COISA":GOTO 40
300 IF LEN(Q$)>1 AND W$="" THEN LET M$="DIGA MAIS DE UMA PALAVRA":GOTO 40
310 IF VB>V AND OB=0 THEN M$="ISTO E TOLICE":GOTO 40
320 IF VB>V AND OB>0 THEN LET M$="FAZER O QUE COM "+W$+"?":GOTO 40
330 IF LEN(Q$)>1 AND OB=0 AND VB<>17 AND VB<>18 THEN LET M$=V$+" O QUE?":GOTO 40
340 IF VB=17 OR VB=18 THEN GOTO 390

```

280-340: Escolhe uma mensagem apropriada.

```
350 LET L=L-1
```

350: Contagem regressiva.

```

360 IF VB>14 THEN GOTO 390
370 ON VB GOSUB 500,570,570,570,570,570,570,570,780,810,810,880,940,1000
380 GOTO 400
390 ON VB-14 GOSUB 1100,1130,1220,1260,1300,1360,1400,1430,1460,1520,1520,1630

```

360-390: Vai para uma sub-rotina que depende do verbo escolhido.

```

400 IF F(14)=0 AND L>0 THEN GOTO 30
410 IF L<1 THEN LET M$="SEU TEMPO ACABOU. O MAXIMIZADOR EXPLODIU!"
420 PRINT M$:PRINT "O JOGO ACABOU":GOSUB 1590:STOP

```

400-420: Fim do jogo.

Escreva I para obter um inventário (lista dos objetos que está carregando).

```

430 FOR I=1 TO W
440 IF W$=Q$(I) THEN LET OB=I
450 IF W$=RIGHT$(Q$(I),LEN(Q$(I))-2) THEN LET OB=I
460 IF LEN(W$)<6 THEN GOTO 480
470 IF RIGHT$(W$,LEN(W$)-5)=Q$(I) THEN LET OB=I
480 NEXT I
490 RETURN

```

430-490: Sub-rotina que verifica se a segunda parte da instrução é o nome de um objeto conhecido.

```

500 PRINT "INVENTARIO":PRINT:PRINT "VOCE ESTA CARREGANDO:"
510 FOR I=1 TO G
520 IF L(1)=25 THEN PRINT Q$(I); " ";
530 NEXT I
540 LET M$="":PRINT
550 GOSUB 1620
560 RETURN

```

500-560: Sub-rotina que mostra na tela uma lista dos objetos que o jogador está carregando.



```

570 LET D=0:IF OB=0 THEN LET D=VB-2
580 IF OB>24 THEN LET D=OB-24
590 IF R=19 AND D=6 THEN LET R=R+1:LET F(13)=1:GOTO 740
600 IF R=20 AND D=6 AND F(9)=0 THEN LET M$="O GATO NAO DEIXA":RETURN
610 IF R=20 AND D=6 AND F(9)=1 THEN LET R=R+1:LET F(13)=1:GOTO 740
620 IF (R=20 OR R=21) AND D=5 THEN LET R=R-1:LET F(13)=1:GOTO 740
630 IF R=2 AND L(1)=25 AND D=3 THEN LET M$=M$:RETURN
640 IF R=7 AND F(3)=0 AND D=2 THEN LET M$="UM RATO FURIOSO IMPEDE A SUA PASSAGEM":RETURN
650 LET F(13)=0:LET RL=LEN(R$(R))
660 FOR I=1 TO RL:LET US=MID$(R$(R),I,1)
670 IF (R=11 AND D=3 AND F(18)=0) THEN LET M$="A PORTA ESTA FECHADA":RETURN
680 IF (R=10 AND D=4 AND F(18)=0) THEN LET M$="A PORTA ESTA FECHADA":RETURN
690 IF (US="N" AND D=1 AND F(13)=0) THEN LET R=R-6:LET F(13)=1
700 IF (US="S" AND D=2 AND F(13)=0) THEN LET R=R+6:LET F(13)=1
710 IF (US="O" AND D=3 AND F(13)=0) THEN LET R=R-1:LET F(13)=1

```

570-770: Sub-rotina que analisa as instruções de direção.

Se o seu micro é um VIC 20, vai precisar de memória adicional.


```

720 IF (US="L" AND D=4 AND F(13)=0) THEN LET R=R+1:LET F(13)=1
730 NEXT I
740 LET MS="OK"
750 IF F(13)=0 THEN LET MS="VOCE NAO PODE IR NESTA DIRECAO"
760 IF D<1 THEN LET MS="IR PARA ONDE?"
770 RETURN

```

Já jogou algumas vezes e não conseguiu chegar ao fim? Dê a você mesmo um pouco mais de tempo aumentando o valor de L na linha 1920.



```

780 LET MS="ESTA TREINANDO PARA A OLIMPIADA?"
790 IF F(10)=1 AND (R=9 OR R=3) THEN LET MS="NAO DA PARA PULAR"
800 RETURN
810 IF OB=8 THEN LET MS=VS+" "+US+M2$+RETURN
820 IF OB=6 THEN LET MS="NAO PODE PEGAR "+US+RETURN
830 IF L(OB)<R THEN LET MS="NAO ESTA AQUI"
840 IF F(OB)<0 THEN LET MS="QUE"+RIGHT$(US,LEN(US)-2)+"?"
850 IF L(OB)=25 THEN LET MS="VOCE JA ESTA COM ELE(A)"
860 IF OB=0 AND L(OB)=R AND F(OB)=0 THEN LET L(OB)=25:LET MS=M3$+US
870 RETURN
880 IF L(OB)<25 THEN RETURN
890 LET MS="NAO PODE!":IF OB<1 THEN RETURN
900 PRINT "COLOCAR A CHAVE ONDE"
910 INPUT US:IF US="" THEN GOSUB 430
920 IF (OB=8 OR US="NO TREM") THEN LET MS=M4$:LET F(15)=1:LET L(1)=R
930 RETURN
940 IF F(11)=0 AND L(7)=25 THEN LET F(11)=1:LET F(3)=0:LET L(3)=3:LET MS=M5$
950 IF F(11)=1 AND L(7)=25 AND R=20 THEN LET F(9)=1:LET MS=M5+M6$
960 IF R=21 AND OB=24 THEN LET F(4)=0:LET MS="ESTA ABERTA"
970 IF OB=16 THEN GOSUB 1000
980 IF OB=10 AND (R=10 OR R=11) THEN LET F(18)=1:MS="ESTA ABERTA":RETURN
990 RETURN
1000 LET MS="NADA DE IMPORTANTE"
1010 IF OB=16 AND R=12 THEN LET F(2)=0:LET MS="E UM MICRO-VIDEOCASSETTE"
1020 IF OB=24 AND R=21 THEN LET MS="HA ALGUMA COISA DENTRO"
1030 IF OB=23 AND R=12 THEN LET MS=M7$
1040 IF OB=7 THEN LET MS="CONTEM UMA MOSCA GIGANTESCA"
1050 IF OB=9 AND R=20 AND F(9)=0 THEN LET MS="ELE MORDE E ARRANHA!"
1060 IF OB=4 THEN GOSUB 1100
1070 IF OB=1 AND (L(1)=25 OR R=16) THEN LET MS="TEM GRAVADO O NUMERO '111'"
1080 IF OB=6 AND (L(6)=25 OR R=17) THEN LET MS="TEM UM GRANDE BOTAO VERMELHO"
1090 RETURN
1100 LET MS=M8$:IF OB=4 AND L(OB)=25 THEN LET MS=M9$
1110 IF OB=4 AND L(OB)<25 THEN LET MS="PRIMEIRO PRECISA PEGA-LO"
1120 RETURN
1130 IF OB<5 THEN LET MS="NAO PODE AMARRAR "+US+RETURN
1140 IF L(5)<25 THEN RETURN
1150 LET US="":PRINT "AMARRAR A LINHA EM QUE"
1160 INPUT US
1170 US=RIGHT$(US,LEN(US)-1)
1180 IF US="" THEN GOSUB 430
1190 LET MS="NAO PODE AMARRAR N"+US
1200 IF OB=13 AND R=9 THEN F(5)=1:LET L(5)=0:LET MS=M8$
1210 RETURN
1220 IF OB=13 OR OB=0 AND R=9 THEN LET MS="E LISA DEMAIS PARA SUBIR"

```

780-1580: Estas são as sub-rotinas para os verbos usados no programa. Procure as linhas com **RETURN** para ver onde uma sub-rotina acaba e a seguinte começa.

A menos que esteja usando um micro TK, pode omitir todos os **LET**.





```

1230 IF R=8 AND F(5)=1 THEN LET R=9:LET F(10)=0:LET MS="OK":RETURN
1240 IF OB=W AND R=19 THEN LET OB=30:GOSUB 570
1250 RETURN
1260 IF R=9 AND F(5)=1 THEN LET R=8:LET F(10)=1:LET MS="OK":RETURN
1270 IF R=9 AND F(5)=0 THEN LET MS="NAO PODE DESCER!"
1280 IF OB=W AND R=20 THEN LET OB=29:GOSUB 570
1290 RETURN
1300 IF OB(>)6 OR L(6)<25 THEN RETURN
1310 LET WS="":PRINT "APONTAR PARA ONDE"
1320 INPUT WS:IF WS="" THEN GOSUB 430
1330 IF OB=22 THEN LET F(6)=1
1340 LET MS="MUITO BEM"
1350 RETURN
1360 IF OB<3 THEN LET MS="QUE BOBAGEM!":RETURN
1370 IF OB=3 AND L(OB)<25 THEN LET MS="PRIMEIRO PRECISA PEGA-LO"
1380 IF OB=3 AND L(OB)=25 THEN LET L(OB)=0:LET MS="NHAM! NHAM!"
1390 RETURN
1400 IF (OB=19 OR OB=17) AND R=24 AND L(1)<25 THEN LET MS=M9$:LET L=L-12:LET F(16)=1
1410 IF (OB=19 OR OB=17) AND R=24 AND L(1)=25 THEN LET F(12)=1:LET MS=MH$
1420 RETURN
1430 IF OB=3 AND R=7 AND L(OB)=25 THEN LET F(3)=1:LET L(OB)=0:LET B=B+1:LET MS=MS
1440 IF L(OB)=25 THEN LET L(OB)=R:LET MS="FEITO"
1450 RETURN
1460 IF R<24 THEN LET MS="NAO HA NENHUM TECLADO AQUI!":RETURN
1470 IF F(12)=0 THEN LET MS="O TERMINAL ESTA TRANCADO":RETURN
1480 LET MS="O TERMINAL REPETE :"+WS
1490 IF OB=18 THEN LET MS="A SENHA ESTA CORRETA":LET F(17)=1
1500 IF OB=20 THEN LET MS="IDENTIFICACAO DO TERMINAL"
1510 RETURN
1520 IF R=14 AND OB=21 AND L(8)=22 THEN LET L(8)=14:LET MS=MB$+MC$:RETURN
1530 IF R=14 AND OB=21 AND L(8)=14 THEN LET L(8)=22:LET MS=MB$+MD$
1540 IF F(15)=1 THEN LET L(1)=L(8)
1550 IF OB=21 AND F(6)=1 AND L(6)=25 AND R=18 AND F(17)=1 THEN LET B=B+10:LET MS=ME$:LET F(14)=1
1560 IF OB=21 AND F(6)=1 AND L(6)=25 AND R=18 AND F(17)=0 THEN LET MS="O MAXIMIZADOR ESTA DESLIGADO"
1570 IF OB=22 THEN LET MS="NAO PODE!"
1580 RETURN

```

Este programa não tem tantas explicações quanto os outros porque isso tiraria a graça do jogo. Depois de jogar algumas vezes, procure descobrir para que servem todas as linhas.



```

1590 LET S=0:FOR I=1 TO 6:IF L(I)=25 THEN S=S+1
1600 NEXT I
1610 LET S=S+8:PRINT "VOCE FEZ ";S;" PONTOS":RETURN
1620 PRINT "APORTE RETURN PARA CONTINUAR":INPUT QS
1630 RETURN

```

1590-1630: Sub-rotina que calcula a contagem de pontos.

```

1640 LET V=25:LET W=31:LET G=8
1650 DIM RS(24):DIM OS(W):DIM HS(V):DIM L(G):DIM F(18)

```

1640-1650: Inicializa variáveis e dimensiona matrizes.

```

1660 DATA "DENTRO DO BURACO DO RATO...ESTA MUITO ESCURO"
1670 DATA "NA ENTRADA DE UM BURACO DE RATO","NA BORDA DE UMA MESA GIGANTESCA"
1680 DATA "EM UM VESTIBULO","EM UM QUARTO COR-DE-ROSA","NA COZINHA"
1690 DATA "NO MEIO DE UM TUNEL ESCURO E MAL-CHEIROSO","PERTO DE UMA ESTRADA DE FERRO"
1700 DATA "PERTO DE UMA ARVORE DE PLASTICO, NA BORDA DE UMA MESA GIGANTESCA"
1710 DATA "DO LADO DE FORA DE UMA CASA DE APARENCIA ESTRANHA"
1720 DATA "EM UMA SALA TODA AMARELA","PERTO DE UM APARELHO DE TV E UM GRAVADOR"
1730 DATA "NO FINAL DE UM TUNEL ESCURO","PERTO DE UMA CAIXA COM UM BOTAO, LIGADA AOS TRILHOS"
1740 DATA "AO LADO DOS TRILHOS","NO FIM DA LINHA...OS TRILHOS DESAPARECEM EM UM BURACO NA PAREDE"

```

1660-1880: Linhas de dados.


```

1750 DATA "PERTO DE UMA PAREDE COM ENORMES TELAS DE TV", "EM PE NA PLATAFORMA DO MAXIMIZADOR"
1760 DATA "EM UMA PRATELEIRA COM ESTRANHOS APARELHOS...HA UMA ESCADA ENCOSTADA NA PRATELEIRA"
1770 DATA "NO MEIO DE UMA ESCADA", "PERTO DE UMA CAIXA DE ASPECTO CURIOSO"
1780 DATA "EM UM BURACO NA PAREDE DO QUAL SAEM TRILHOS", "DENTRO DE UM LABORATORIO"
1790 DATA "PERTO DE UM TERMINAL DE COMPUTADOR TRANCADO A CHAVE"
1800 DATA 16,12,3,21,5,17,19,14
1810 DATA "I", "IR", "M", "S", "O", "L", "C", "B", "PULAR", "PEGAR", "LEVAR", "COLOCAR", "ABRIR", "EXAMINAR"
1820 DATA "LER", "AMARRAR", "SUBIR", "DESCER", "APONTAR", "COMER", "DESTRANCAR", "LARGAR", "ESCREVER"
1830 DATA "APERTAR", "LIGAR", "SL", "SO", "SL", "SOL", "OL", "SO", "NS", "NS", "NL", "NLO", "LO"
1840 DATA "NO", "NS", "NL", "LO", "O", "S", "S", "NB", "CB", "LC", "LO", "NLO", "NO", "A CHAVE", "A FITA"
1850 DATA "O QUEIJO", "O PAPEL", "A LINHA", "O CONTROLE REMOTO", "A GARRAFA", "O TREM", "O GATO"
1860 DATA "A PORTA", "A MESA", "O TUNEL", "A ARVORE", "O BURACO", "O RATO", "O GRAVADOR"
1870 DATA "O COMPUTADOR", "COLIBRI", "O TERMINAL", "111", "O BOTAO", "O MAXIMIZADOR", "A TV", "A CAIXA"
1880 DATA "PARA NORTE", "PARA SUL", "PARA OESTE", "PARA LESTE", "PARA CIMA", "PARA BAIXO", "A ESCADA"

```

```

1890 FOR I=1 TO 24:READ D$:NEXT I:FOR I=1 TO 6:READ L(I):NEXT I
1900 FOR I=1 TO V:READ H$(I):NEXT I:FOR I=1 TO 24:READ R$(I):NEXT I
1910 FOR I=1 TO W:READ O$(I):NEXT I:LET F(10)=1:LET F(4)=1:LET F(2)=1

```

1890-1910: Leitura de dados.

```

1920 LET R=11:LET B=8:LET L=100:LET M$="VOCE ACORDA...":LET MH$="O TERMINAL IMPRIME:(SENHA)?"
1930 LET M1$="A CHAVE NAO PASSA":LET M2$="HEIN? NAO PODE!"
1940 LET M3$="VOCE ESTA COM ":LET M4$="MUITO BEM!":LET M5$="NADA DE IMPORTANTE"
1950 LET M5$="UMA MOSCA ENORME SAI VOANDO":LET M6$="O TREM "
1960 LET M7$="E APENAS UMA CAIXA COM UMA FOTO":LET M9$="ALARME LIGADO!"
1970 LET M8$="ESTA BEM AMARRADA":LET M6$=" E O GATO SAI CORRENDO ATRAS!"
1980 LET M9$="O RATO SAI CORRENDO COM ELE.":LET M5$="SENHA DO TERMINAL="+O$(18)
1990 LET M5$="CHEGA E PARA AQUI":LET M5$="SE AFASTA E ENTRA EM UM TUNEL"
2000 LET M5$="O RATO DO MAXIMIZADOR FUNCIONOU.VOCE VOLTOU AO TAMANHO NORMAL":RETURN

```

1920-2000: Mensagens.

Modificações

- ◆◆◆◆◆ Suprima a linha 10
- 40 Substitua CLS por HOME
- ▲ 40 Substitua CLS por PRINT CHR\$(147)

...PARA OS MICROS TK

```

140 Substitua PRINT MID$(R$(R),I,1) por PRINT R$(R,I)
200 FOR I=1 TO LI-1
220 IF O$(I)=" " AND V$="" THEN LET V$=O$(I) TO I-1)
210 IF O$(I+1)<>" " AND V$<>"" THEN LET W$=O$(I+1 TO I):LET I=LI-1
245 IF LEN(V$)<7 THEN LET V$=V$+" ":GOTO 245
370 GOSUB 500*(VB=1)+570*(VB=1) AND VB(9)+780*(VB=9)+810*(VB=10 OR VB=11)
+880*(VB=12)+940*(VB=13)+1000*(VB=14)
390 GOSUB 1100*(VB=15)+1130*(VB=16)+1220*(VB=17)+1260*(VB=18)+1300*(VB=19)+1360*(VB=20)
+1400*(VB=21)+1430*(VB=22)+1460*(VB=23)+1520*(VB=24 OR VB=25)+1630*(VB=26)
430 IF LEN(W$)<14 THEN LET W$=W$+" ":GOTO 430
435 FOR I=1 TO W
660 FOR I=1 TO RL:LET U$=R$(R,I)
1650 DIM R$(24,4):DIM O$(W,14):DIM H$(V,7):DIM L(6):DIM F(18)
1920 LET R=11:LET B=8:LET L=100:LET M$="VOCE ACORDA..."
2000 RETURN

```

Linhas 630,810,860,920,940,950,1030,1100,1200,1400,1410,1430,1520,1530,1550

Substitua M1\$ a MH\$ nestas linhas pelo texto das linhas 1920 a 2000

Exemplo: 630 IF R=2 AND L(1)=25 AND D=3 THEN LET M\$="A CHAVE NAO PASSA":RETURN

Suprima as linhas 1920 a 1990

NOTA

IMPORTANTE: Se o seu micro é o Commodore 64 ou o VIC 20, elimine os espaços e os LET das linhas mais compridas, caso contrário o computador não as aceitará.



Respostas dos problemas

Aqui estão as respostas de alguns dos problemas propostos.

Os Crânios da Pirâmide

Para aumentar o número de crânios, substitua o 5 da linha 260 por um número maior.

Os limites do número que o jogador deve obter são fixados na linha 20. No programa original, o número está entre 10 e 39. Se você mudar o 30 para 40, obterá um número entre 10 e 49. Se também mudar o 10 para 20, o número estará entre 20 e 59.



O Desafio dos Monstros

Para aumentar o número de *rounds*, substitua o número 11 nas linhas 260 e 550 por um número maior (o número deve ser o mesmo nas duas linhas).

Se você aumentar o número de *rounds*, sentirá necessidade de recorrer mais vezes ao Botão de Pânico. Por isso, é aconselhável substituir o número 4 na linha 440 por um número maior.

Tubarão

Para fazer o caçador se mover mais freqüentemente, mude o valor de U na linha 30 para um número um pouco menor, como 0,8, por exemplo.

Receita de Bruxa

Quando você perde o programa vai para a linha 240 ou 250 e imprime uma mensagem. Mude essas linhas para o que você quiser.

Nota especial para os usuários do TK-2000

Se o seu micro é um TK-2000, use as modificações indicadas para a família Apple, com uma única exceção: se a modificação for do tipo

```
nnn A$="":IF PEEK(-16384))127 THEN GET A$
```

onde nnn é o número da linha e A\$ é o nome de uma variável, substitua esta linha por:

```
nnn A$="":CALL -16397:IF PEEK(-16385))0 THEN GET A$
```

e acrescente as seguintes linhas:

```
1 DATA 169,255,141,0,192,77
```

```
2 DATA 16,192,141,255,191,96
```

```
3 FOR I=-16397 TO -16386
```

```
4 READ A:POKE I,A:NEXT I
```

Nota para os usuários do MSX: Em todos os programas, acrescente a linha:

```
5 R=RND(-TIME)
```

Nota para os usuários do PC: Em todos os programas, acrescente a linha:

```
5 RANDOMIZE VAL(RIGHT$(TIME$,2))
```


Guias Práticos de Microcomputadores



TUDO O QUE VOCÊ PRECISA SABER SOBRE MICROCOMPUTADORES

e as maravilhas que eles podem fazer



NOVOS * COLORIDOS * FASCINANTES

923607

LUTECIA